LCD Serial Backpack® and 2x16 Serial LCDs

The LCD Serial Backpack® is a daughterboard that attaches to standard character LCD modules. It receives data serially and displays it on the LCD. The Backpack supports any alphanumeric LCD up to 80 screen characters (e.g. 4 lines by 20 characters). It accepts serial data at 2400 or 9600 baud (switch selectable). It is sold by itself, or preinstalled to high-quality 2x16 LCD modules.

The Backpack has two modes: *text* and *instruction*. It defaults to text mode; any data sent to the Backpack is displayed on the screen. Send the string "HELLO" and "HELLO" appears on the LCD. To distinguish text from instructions (e.g., clear screen, position cursor, etc.), the Backpack looks for an instruction prefix (ASCII 254). The byte following prefix is treated as an instruction. After the instruction code, the Backpack returns to text mode.

Table of Contents

Disclaimer of Liability

Scott Edwards Electronics, Inc. is not responsible for any special, incidental, or consequential damages resulting from any breach of warranty, or under any legal theory, including lost profits, downtime, goodwill, damage to or replacement of equipment or property, and any costs or recovering, reprogramming, or reproducing of data associated with the use of the hardware or software described herein.

Warranty

Scott Edwards Electronics, Inc. warrants this product against defects in materials and workmanship for a period of 90 days. If you discover a defect, we will, at our option, repair, replace, or refund the purchase price. Return the product with a description of the problem. We will return your product or its replacement via standard shipping. Expedited shipping is available at the customer's expense.

• Note: Physically abusing the module, or attempting to repair or modify it, voids this warranty.

Trademarks and Copyrights

LCD Serial Backpack® is a registered trademark of Scott Edwards Electronics, Inc.; BASIC Stamp® is a registered trademark of Parallax Inc. All trademarked names referenced herein are the property of their respective holders.

Description of the Backpack Board, Installation to Typical LCDs

Figure 1 shows the major features and configuration options of the Backpack circuit board. Figure 2 shows how to mount it to typical LCD modules.



Do not reverse +5V and GND, even momentarily. Reversed power will destroy the electronics.
Do not exceed 5.5V into +5V. Overvoltage will damage the unit or shorten its life.

Figure 1. The LCD Serial Backpack® circuit board.



Figure 2. Mounting the Backpack to an LCD.

Quick Checkout and Contrast Adjustment

You can test the Backpack and LCD for proper operation without a computer/serial port. Temporarily connect the serial input to one of the +5 terminals of J1, then connect power to +5 and GND. Backpacks with pcb revs 5 and 6 made before March 1998 will display rows of triangles across the display. Later rev 6 and higher units (ones shipped with this manual) will display a text message. Units earlier than rev 5 will produce unpredictable results with this test, so they should be tested with an actual serial input.

The contrast control is usually set fully clockwise at the factory. This setting may be too dark with some LCDs. Use a small, flat-blade screwdriver to adjust the contrast.

Configuration

Display Lines: Set the LINES switch (figure 1) down for 1-line displays; up for 2- or 4-line displays. Note that most 1x16 displays behave like 2x8 displays (with the two lines side by side). If you have a 1x16 LCD and cannot see the second eight characters, power down and switch to the 2-/4-line setting.

Baud Rate: Set the BPS switch down for 2400 baud; up for 9600. At either rate, the serial characteristics are no parity, 8 data bits, 1 stop bit.

NOTE: The Backpack reads the configuration switches only at startup. Change switch settings only with the power off.

Hookup for Use

Figure 3 shows how to connect the LCD Serial Backpack® to PCs and BASIC Stamp computers in order to run the example programs presented later in this manual. Refer to figure 1 or the markings on the Backpack for the locations of +5, GND and SER(ial in).

The 5-pin connector on the Backpack has two extra pins for +5 and GND. These pins are arranged in a *pallindrome* layout. If you make a matching 5-pin connector, the connections will always line up properly regardless of connector orientation. Suitable female crimp sockets for making connectors are available from Jameco (www.jameco.com, PN: 100765). Digi-Key carries ready-made 5-pin socket and flex-cable assemblies in varying lengths (www.digikey.com, PN: A9BAG-0506F-ND).



Figure 3. Connecting to PC serial port, BASIC Stamps.

Basic Operation

Once the Backpack is properly connected and configured to match the baud rate of the computer/program that will be talking to it serially, data sent to the Backpack will appear on the display. For example, if you send "Hello" to the Backpack, "Hello" appears on the display. The cursor (printing position) automatically advances from left to right.

You can also send instructions to the LCD via the Backpack. To tell the Backpack that a particular byte is an instruction and not text, precede it with the *instruction prefix* character, ASCII 254 (0FE hex, 11111110 binary). The Backpack treats the byte immediately after the prefix as an instruction, then automatically returns to data mode.

An example: The clear-screen instruction is ASCII 1. To clear the screen, send <254><1> (where the <> symbols mean single bytes set to these values, not text as typed from the keyboard). Table 1 lists the LCD instructions.

Table 1. LCD Instructions

Instruction/Action	Code
⧗ Clear Screen	1
Scroll display one character left (all lines)	24
Scroll display one character right (all lines)	28
⧗ Home (move cursor to top/left character position)	2
Move cursor one character left	16
Move cursor one character right	20
Turn on visible underline cursor	14
Turn on visible blinking-block cursor	13
Make cursor invisible	12
Blank the display (without clearing)	8
Restore the display (with cursor hidden)	12
Set cursor position (DDRAM address)	128 + addr
Set pointer in character-generator (CG) RAM	64 + addr

⧗ *These instructions take more than 1 ms for the LCD to execute. Pause at least 1 ms before sending additional data after these instructions. At 2400 bps, no pause is ever needed.*

Advanced LCD users: These are the actual LCD instruction codes. When the Backpack receives the instruction prefix, it clears the register select (RS) bit. The next received byte is written to the LCD with RS low. After writing that byte, it returns RS high (normal data mode). This approach means that you can send any valid LCD instruction through the Backpack. Since the Backpack intializes the LCD at startup, you should not send any initialization instructions. If you did not understand the preceding note, don't worry. You don't need to know any of this to use an LCD with the Backpack. The info may be handy to advanced users adapting programs written to drive the Backpack to applications that drive the LCD directly through the parallel bus.

Positioning the Cursor

You can position the cursor anywhere on the screen by sending the proper instruction. The arrangement of character positions depends on the type of LCD, as shown in figure 4 below.

one-line LCDs
(other than 1x16; set LINES switch to 1)

character position	0	1	2	3	4	5	6	•••
set-position instruction	128	129	130	131	132	133	134	•••

1x16 LCDs
(set LINES switch to 2/4)

character position	0	1	2	•••	8	9	10	•••		
set-position instruction	128	129	130	•••	192	193	194	•••		

two-line LCDs
(set LINES switch to 2/4)

	0	1	2	3	4	5	6	•••
line 1	128	129	130	131	132	133	134	•••
line 2	192	193	194	195	196	197	198	•••

four-line LCDs
(set LINES switch to 2/4)

	0	1	2	3	4	5	6	•••
line 1	128	129	130	131	132	133	134	•••
line 2	192	193	194	195	196	197	198	•••
line 3	148	149	150	151	152	153	154	•••
line 4	212	213	214	215	216	217	218	•••

To position the cursor, send the instruction-prefix byte, ASCII 254, followed by the set-position byte value. For example, to move to line 2, character 4, send <254><196>. Note: <#> means a byte set to the value #. See program examples.

Figure 4. Cursor positioning.

Defining Custom Symbols

Most of the LCD characters (figure 6) cannot be changed because they are stored in ROM. However, the first eight symbols, corresponding to ASCII 0 through 7, are stored in RAM. By writing new values to the character-generator (CG) RAM, you can alter these characters. Changing a symbol is easy; just point to the beginning of the symbol's RAM location, then write eight bytes whose bits form the desired pattern. Then position the cursor onto the screen. See figure 5.

Manipulating custom characters allows you to create all kinds of special effects, including simple animations. See the Backpack application notes at `www.seetron.com` for more examples.

Bitmap Layout



Byte Values	
binary	decimal
xxx00000	0
xxx00100	4
xxx00010	2
xxx11111	31
xxx00010	2
xxx00100	4
xxx00000	0
xxx00000	0

Symbol Locations

ASCII Code	Base Address
0	64
1	72
2	80
3	88
4	96
5	104
6	112
7	120

Example: Loading the Symbol

To load the arrow shown at left to symbol 3, a program would send the following bytes to the LCD:

```
<254><88>          ' Point to symbol 3
<0><4><2><31>      ' Send the bit
<2><4><0><0>       ' pattern
<254><128>         ' Move cursor back
                   ' to the screen
```

<#> means a byte set to the value #. See the program examples.

Figure 5. Defining custom symbols.



NOTE: Custom characters occupy ASCII 0—7
Backpack loads patterns shown at startup.
ASCII 8—31 repeat the custom characters
ASCII 128–160 are blanks

To find the ASCII code for a given character, add the row and column numbers. For example, capital D is in the column marked 64 in row 4, so its ASCII code is 68. Use the reverse procedure to determine the symbol for a given code. For example, ASCII code 244 produces the symbol Ω, found at colum 240, row 4.

Figure 6. LCD character set.

Program Examples

Any computer/programming language that can produce serial output (2400 or 9600 bps, N81) can talk to the LCD Serial Backpack®. The examples here are in BASIC, chosen because of its popularity and readability. Don't be put off by the size of the programs—they are mostly comments.

These examples are meant to illustrate only the fundamentals. More elaborate examples are available from www.seetron.com.


```
' Program: BPKDEMO.BAS
' This program demonstrates fundamental techniques of driving the
' LCD Serial Backpack (R) in BASIC (compatible with QBASIC, Quick BASIC,
' First BASIC, and Power BASIC). First BASIC, an excellent shareware
' compiler, is available from www.powerbasic.com.

' Start by defining some useful constants for the Backpack
I = 254                 ' Instruction prefix value.
CLR = 1                 ' LCD clear-screen instruction.
LINE1 = 128             ' Address of first character of 1st line.
LINE2 = 192             ' Address of first character of 2nd line.

' Open the serial port (com1) for output at 9600 baud. Make sure Backpack
' is also set for 9600. Turn off all handshaking (CD, CS, DS) by
' setting to zero (0).
OPEN "COM1:9600,N,8,1,CD0,CS0,DS0" FOR OUTPUT AS #1

' Once the port is open, we can print to it. Start by clearing the screen
' in case there's text left from a previous run of this program. Note that
' at 9600 baud, you need a delay after clearing the screen. To create a
' delay, you can use a timing instruction such as First/Power BASIC's
' DELAY or QBASIC's SLEEP, or you can send an unnecessary instruction,
' such as <254><128>. That sequence moves the cursor to the beginning of
' line 1, which is where it already is, thanks to clear-screen.
PRINT #1, CHR$(I); CHR$(CLR);   ' Send <254><1> to clear screen.
PRINT #1, CHR$(I); CHR$(LINE1); ' Time delay (for 9600 baud).

' Now print some text. PRINT statements should end with ; to
' prevent unnecessary carriage return/line feeds (which the Backpack
' doesn't understand, and displays as junk characters).
PRINT #1, "Hello world!";

' Positioning the cursor requires sending the instruction prefix (ASCII
' 254, which we've assigned the name "I") followed by an address. We've
' assigned names to ASCII 128 (1st character of line 1) and 192 (1st
' character of line 2). We'll position the cursor to the start of
' line 2 and print some more.
PRINT #1, CHR$(I); CHR$(LINE2); "press return";

' Wait for a keypress (at PC) before continuing.
CLS : INPUT "PRESS RETURN TO CONTINUE", X$

' Now we'll simulate a common application by printing a label on the
' screen, then updating some data by positioning the cursor. Each
' time you press return on the PC, the program will add 1 to the
' count and update the value on the screen. Notice that to position
' the cursor at character 6 of line 1, we give the position value of
' LINE1 + 6. This is easier to read than 134, which is the address of
' line 1, character 6. Also note that we print several spaces after
' the number. It's not needed here, but in programs in which a number
' could be _lower_ than the previously displayed value, the spaces
' would erase any leftover digits.
PRINT #1, CHR$(I); CHR$(CLR); CHR$(I); CHR$(LINE1);
PRINT #1, "Count: "; CHR$(I); CHR$(LINE2); "press return";
theCount = 0
Again:
  PRINT #1, CHR$(I); CHR$(LINE1 + 6); theCount; "   ";
  INPUT "ENTER TO CONTINUE, Q-ENTER TO QUIT ", X$
  IF UCASE$(X$) = "Q" THEN END
  theCount = theCount + 1
GOTO Again
```

Quickie Program

Want fast results? All you need are these two lines of code. The first opens the serial port for output; the second 'prints' text to it.Type and run these lines in QBASIC or Power BASIC and "Hello World!" appears on the display.

```
' Program: BPKDEMS1.BAS
' This program demonstrates fundamental techniques of driving the
' LCD Serial Backpack (R) in BASIC from the BASIC Stamp I. It assumes
' that the Backpack is connected to I/O pin 0 of the Stamp, and
' that it is set for 2400 bps.

' Start by defining some useful constants for the Backpack.
SYMBOL I       =      254     ' Instruction prefix value.
SYMBOL CLR     =      1       ' LCD clear-screen instruction.
SYMBOL LINE2   =      192     ' Address of 1st char of 2nd line.
SYMBOL L1_C7   =      135     ' Address of line 1, character 7.

' Now clear the screen in case there's text left from a previous
' run of the program. Note that there's a 1-second PAUSE prior to
' sending any data to the Backpack. This gives the Backpack plenty
' of time to initialize the LCD after power up.
pause 1000
serout 0,n2400,(I,CLR)        ' Clear the LCD screen.
serout 0,n2400,("Hello World!")       ' Print message.

' Positioning the cursor requires sending the instruction prefix (ASCII
' 254, assigned the symbol "I") followed by an address.
serout 0,n2400,(I,LINE2,"..line 2")   ' Move to line 2 and print.

' Now we'll simulate a common application by printing a label on the
' screen and updating some data by positioning the cursor.
pause 2000                    ' Wait 2 secs.
serout 0,n2400,(I,CLR)        ' Clear the LCD screen.
serout 0,n2400,("Count:")     ' Print the label.
Again:
  serout 0,n2400,(I,L1_C7)    ' Move to line 1, character 7.
  serout 0,n2400,(#b2,"  ")   ' Print value of b2 followed by 2 spaces.
  b2 = b2+1                   ' Increment b2.
  pause 200                   ' Slow the loop down.
goto Again                    ' Repeat endlessly.
```

BASIC STAMP I

Quickie Program

Type and run just this line in PBASIC 1 and "Hello World!" appears on the display.

```
' Program: BPKDEMS2.BS2
' This program demonstrates fundamental techniques of driving the
' LCD Serial Backpack (R) in BASIC from the BASIC Stamp II. It assumes
' that the Backpack is connected to I/O pin P0 of the Stamp, and
' that it is set for 9600 bps.

' Start by defining some useful constants for the Backpack.
N9600   con     $4054   ' Baudmode for 9600 bps inverted.
I       con     254     ' Instruction prefix value.
CLR     con     1       ' LCD clear-screen instruction.
LINE2   con     192     ' Address of 1st char of 2nd line.
L1_C7   con     135     ' Address of line 1, character 7.

' Now clear the screen in case there's text left from a previous
' run of the program. Note that there's a 1-second PAUSE prior to
' sending any data to the Backpack. This gives the Backpack plenty
' of time to initialize the LCD after power up.
pause 1000
serout 0,n9600,[I,CLR]        ' Clear the LCD screen.
pause 1
serout 0,n9600,["Hello World!"]       ' Print message.

' Positioning the cursor requires sending the instruction prefix (ASCII
' 254, assigned the symbol "I") followed by an address.
serout 0,n9600,[I,LINE2,"..line 2"]   ' Move to line 2 and print.

' Now we'll simulate a common application by printing a label on the
' screen and updating some data by positioning the cursor. N' serout 0,n9600,[I,L1_C7,DEC
b2,"  "].
pause 2000                    ' Wait 2 secs.
serout 0,n9600,[I,CLR]        ' Clear the LCD screen.
pause 1
serout 0,n9600,["Count:"]     ' Print the label.
Again:
  serout 0,n9600,[I,L1_C7]    ' Move to line 1, character 7.
  serout 0,n9600,[DEC b2,"  "]       ' Print value of b2 followed by 2 spaces.
  b2 = b2+1                   ' Increment b2.
  pause 200                   ' Slow the loop down.
goto Again                    ' Repeat endlessly.
```

BASIC STAMP II

Quickie Program

Type and run just these lines in PBASIC 2 and "Hello World!" appears on the display.

Dimensions and Backlight Hookup

The drawing below gives the primary dimensions of the 2x16 LCDs sold with our BPK-216N and BPK-216L products. Three LCD manufacturers are listed, Truly, Data Vision (DV), and Powertip (PT), since we have at times used LCDs from each of these companies. Powertip is our primary supplier.

		Truly	DV	PT
A	y offset pcb edge to hole ctr	2.50	2.50	2.50
B	y pcb height	36.00	36.00	36.00
C	y hole spacing	31.00	31.00	31.00
D	y screen opening	13.80	16.10	16.20
E	y character size	4.35	4.89	5.94
F	x character size	2.95	2.78	2.95
G	x offset pcb edge to hole ctr	2.50	2.50	3.55
H	x screen frame	73.70	71.20	71.00
I	x screen opening	64.50	62.50	66.00
J	x hole spacing	75.00	75.00	75.00
K	x pcb width	80.00	80.00	80.00
L	y frame height	24.80	26.20	25.00
-	mounting hole diameter	2.50	2.50	2.50
-	frame depth, non-backlit	5.00	4.70	5.00
-	frame depth, LED-backlit	8.50	9.40	8.50

- All dimensions in mm.
- Worst-case tolerance for any dimension is ±0.50mm.
- Maximum depth (from front of screen frame to tips of interface header posts) is 26mm.
- Dimensions based on information provided by manufacturers—subject to change without notice.



LED Backlight

Locate the terminals marked A and K next to the LCD screen and connect to power as shown.



Figure 7. Physical dimensions and backlight connection.

Revision History, Basic Specifications

As of 03/98, the LCD Serial Backpack has been in continuous production for more than four years. Table 2 summarizes the revisions. Note that "pcb update" does not necessarily indicate a change in the board layout, but a change in pcb manufacturing facility or generation of fresh tooling.

Table 2. Revisions

REV 1 ..... 2400 baud only
REV 2 ..... 9600-baud capability added
REV 3 ..... minor pcb updates
REV 3A ..... firmware changed to use instruction prefix
REV 4 ..... minor pcb updates
REV 5 ..... jumpers changed to configuration switches
REV 6 ..... minor pcb updates; later rev 6 models display test msg
..... when SER pin is connected to +5V throughout startup

Table 3. Basic Specifications

Power requirements (incl. typical LCD) ..... 4.8 to 5.5 Vdc @ 3 mA
User connector ..... Five 0.025" pins on 0.10" centers
Connector pinout ..... +5 GND SER GND +5
Serial input ..... RS-232, or inverted TTL/CMOS, 2400 or 9600, N81
Operating temperature ..... 0° to 50°C (32° to 122°F)
Storage temperature ..... –10° to 60°C (14° to 140°F)
LCD type (BPK-216x series) ..... Supertwist (STN), yellow-green
Optimum viewing direction ..... 6 o'clock